

TTX-generator for the SAA7182/3 (EURO-DENC)

Author: John McNally

DESCRIPTION

The Teletext Module is used to provide digital data to the TTX pin of the SAA7182/3 for encoding onto its analog outputs in the correct form for further decoding inside a television.

A microcontroller reads the **TTX** data from an EPROM which contains six pages of Teletext data. It is read out of the EPROM and into a FIFO line by line as each line is requested. The 8 bit data is converted to serial form, under control of the microcontroller and a multiplexer, before being stored in the FIFO. The write clock, write enable and **RSTW** are controlled by the microcontroller. The function of the FIFO is to decouple the slower write bit stream of the microcontroller from the fast 27MHz bit stream required by the DENC. The **RSTR** signal for the FIFO is signal **VS** from the DENC.

Signal **TTXREQ** from the DENC starts the process. When it goes high the data in the FIFO is read into the TTX pin of the DENC at the appropriate bit rate for encoding. When **TTXREQ** goes low it causes an interrupt. The microcontroller will then reset the write pointer and write the next line into the FIFO.

When Teletext is active the DENC will start to output bits 10.2 μ S after the start of the sync pulse regardless of the data on the **TTX** input. Therefore it is up to the user to program **TTXREQ** to be active at a point that will allow the module to respond and send out the data such that the first bit encoded is the first bit of the run in code. The pipeline delay in the DENC is about 70 LLC clocks. The easiest method to get the correct position of **TTXREQ** is by using a scope with two inputs. One input is connected to **TTXREQ** which is used as the trigger. The other is connected to **Pin 73 CVBS** on the DENC. Register **TTXHS** can then be adjusted so that the run in and framing codes can be seen as a stationary sine wave on the scope. The remainder of the line will be continuously changing as the data changes. Register **TTXHE** can then be adjusted to give the correct length of **TTXREQ**.

PARTS LIST

TYPE	NUMBER
Semiconductors	
IC 74151 (Multiplexer)	1
TMS4C1070 (FIFO)	1
74HCT08 (AND-Gate)	1
27256 (256k PROM)	1
87C654 (microcontroller)	1
Capacitors	
100 nF	5
33 pF	2
2.2 μ F	1
10 μ F	1
Resistors	
10 k Ω	1
8 $\times$ 1 k Ω (Array)	2




Figure 1. Block diagram of the TTX Module

TTX-generator for the SAA7182/3 (EURO-DENC)

ASSEMBLER CODE FOR THE 80C654

```

$TITLE(U.Bergmann TTX-MODUL fuer 87654)
; ENGLISH TRANSLATION BY JOHN MCNALLY
$DATE(08.05.1995)
;OPERATION:-
;THE MICRO INITIALIZES ITS PORTS AND REGISTERS THEN LOOPS FOREVER.
;WHEN AN INTERRUPT COMES FROM TTXREQ GOING LOW THE MICRO DISABLES FIFO
;READS THEN WRITES A LINE OF TEXT DATA INTO THE FIFO. THE DATA CONTAINS
;EVERYTHING REQUIRED BY A TEXT DECODER INCLUDING RUN IN AND FRAME BYTES.
;THE DATA IS WRITTEN ACCORDING TO THE EURO DENC PROTOCOL REGARDING THE
;NUMBER OF CLOCKS PER TTX BIT. IE THREE OR FOUR. THE DATA FROM THE
;EPROM IS SERIALIZED USING AN EIGHT TO ONE MUX CONTROLLED BY THE MICRO.
;THEN WHEN TTXREQ IS HIGH, THE DATA FOR A LINE IS READ OUT FROM
;THE FIFO WITH NO INTERVENTION FROM THE MICRO. THE READ POINTER IS RESET
;EVERY FIELD BY VS.
;*****
;REGISTER DEFINITION
;*****
RSTW      BIT    P1.3          ; RESET FIFO WRITE POINTER
SWCK      BIT    P1.4          ; FIFO WRITE CLOCK
READ      BIT    P1.5          ; FIFO READ ENABLE = READ*TTXREQ
L_ADR     EQU    DPL           ; EPROM LOW ADDRESS BYTE
H_ADR     EQU    DPH           ; EPROM HIGH ADDRESS BYTE
BYTE_NO   EQU    R0            ; TTX BYTE NUMBER. 1 TO 37
BYTES     EQU    R2            ; NO. OF BYTES TO SEND
NBR_BYTE  EQU    40H           ; ACTIVE BYTES PER LINE
NBR_BIT   EQU    08H           ; MUX BIT COUNT
NBR_OUT   EQU    04H           ; MAXIMUM NO OF COUNTS PER BIT
ADR_START EQU    0200H         ; ROM START ADDRESS
L_ADR_END EQU    00H           ; LAST LOW ADDRESS OF ROM
H_ADR_END EQU    2CH           ; LAST HIGH ADDRESS OF ROM
;*****
;RESET AND INTERRUPT CALLS
;*****
ORG 0000H
RES:       LJMP MAIN           ; JUMP TO MAIN PROGRAM AFTER RESET
ORG 0003H
IN0:       LJMP INTERRUPT      ; JUMP TO THE INTERRUPT ROUTINE
; INT0 IS CONNECTED TO TTXREQ

ORG 000BH
TIF0:      RETI
ORG 0013H
IN1:       RETI
ORG 001BH
TIF1:      RETI
ORG 0023H
RIT1:      RETI
ORG 0100H

```

TTX-generator for the SAA7182/3 (EURO-DENC)

```

;*****
;
;MAIN PROGRAM
;*****
MAIN:      CALL INIT          ; INIT
           SETB IE.7          ; ENABLE INTERRUPTS
           SETB IE.0          ; ENABLE INT0
LOOP:      JMP LOOP           ; WAIT FOR AN INTERRUPT
;*****
;SUB ROUTINES
;*****
;*****
;REGISTER- AND BIT INITIALIZATION
;*****
INIT:      MOV P1,#00H        ; PORT1 = 00H
           MOV P3,#00H        ; PORT3 = 00H
           MOV DPTR,#ADR_START ; RESET THE VARIABLES
           MOV BYTES,#00H     ; BYTE COUNTER = 00H
           MOV BYTE_NO,#01H   ; BYTE_NO = FIRST
           SETB IP.0          ; SET INTERRUPT PRIORITY LEVEL
           NOP
           SETB TCON.0        ; ACTIVE LOW EDGE
                               ; TRIGGERED INTERRUPT
           NOP
           RET
;*****
;WRITE A TTX-LINE
;*****
WRT_TXTLINE:
           MOV BYTES,#NBR_BYTE ; LOAD BYTE COUNTER
           MOV BYTE_NO,#01H    ; BYTE_NO STARTS AT ONE
TXT_LOOP:
           MOV P0,L_ADR
           MOV P2,H_ADR        ; ADDRESS THE EPROM
MUX:
           MOV R6,#00H
MUX_LOOP:
           MOV P1,R6           ; PUT EACH BIT IN THE FIFO
VALUE:
           MOV R5,#NBR_OUT
HOWMANY:
           MOV A,#0AH
           XRL A,BYTE_NO
           JZ THREETIMES
           MOV A,#13H          ; ALGORITHM TO DETERMINE
                               ; WHETHER EACH BIT IS
                               ; CLOCKED THREE OR FOUR TIMES
           XRL A,BYTE_NO        ; INTO THE FIFO
           JZ THREETIMES
           MOV A,#1CH
           XRL A,BYTE_NO
           JZ THREETIMES
           MOV A,#25H
           XRL A,BYTE_NO

```

TTX-generator for the SAA7182/3 (EURO-DENC)

```

                JZ  FULLBYTE_NO
                SJMP BIT_LOOP
FULLBYTE_NO:
                MOV BYTE_NO, #00H
THREETIMES:
                DEC  R5
BIT_LOOP:
                NOP
BIT_LOOP2:
                SETB SWCK          ; SET THE CLOCK-BIT
                NOP                ; WAIT
                CLR  SWCK          ; RESET THE CLOCK-BIT
                NOP                ; WAIT
                DJNZ R5, BIT_LOOP2
                INC  BYTE_NO        ; BLOCK = 36+1
                INC  R6             ; INC MUX ADDRESS
                MOV  A, #NBR_BIT
                XRL  A, R6
                JNZ  MUX_LOOP
                INC  DPTR
                DJNZ BYTES, TXT_LOOP
                RET
;*****
;
; INTERRUPT 0 ROUTINE
;*****
;
INTERRUPT:
                CLR  IE.7           ; DISABLE INTERRUPTS
RESET_FIFO:
                SETB RSTW          ; RESET FIFO WRITE POINTER
                SETB SWCK          ; SET THE CLOCK-BIT
                NOP
                CLR  SWCK          ; RESET THE CLOCK-BIT
                CLR  RSTW
DIS_READROM:
                CLR  READ          ; DISABLE FIFO READ
                CALL WRT_TXTLINE   ; WRITE A TEXT LINE
EN_READROM:
                SETB READ          ; ENABLE FIFO READ
ADR_CHECK:
                MOV  A, #H_ADR_END
                XRL  A, H_ADR
                JNZ  CHK_END
                MOV  A, #L_ADR_END
                XRL  A, L_ADR
                JNZ  CHK_END
                MOV  DPTR, #ADR_START ; RELOAD START ADDRESS
CHK_END:
                SETB IE.7          ; ENABLE INTERRUPTS
                SETB IE.0          ; ENABLE INTO
                RETI
;*****
;
END

```

TTX-generator for the SAA7182/3 (EURO-DENC)

POSSIBLE REGISTER SETTINGS FOR THE SAA7182/83 IN TELETEXT MODE

REG 3a = c7	REG 60 = 70
REG 42 = 6b	REG 61 = 6
REG 43 = 0	REG 62 = 4b
REG 44 = 0	REG 63 = cb
REG 45 = 52	REG 64 = 8a
REG 46 = 90	REG 65 = 9
REG 47 = 12	REG 66 = 2a
REG 48 = 2a	REG 67 = 55
REG 49 = 26	REG 68 = 56
REG 4a = 90	REG 69 = 67
REG 4b = 11	REG 6a = 58
REG 4c = b6	REG 6b = 12
REG 4d = a2	REG 6c = 2a
REG 4e = ea	REG 6d = 1
REG 4f = 4a	REG 6e = 20
REG 50 = 5e	REG 6f = 31
REG 51 = d1	REG 70 = 80
REG 52 = da	REG 71 = e8
REG 53 = 70	REG 72 = 10
REG 54 = a9	REG 73 = 3e
REG 55 = 70	REG 74 = b8
REG 56 = ee	REG 75 = 61
REG 57 = 90	REG 76 = 15
REG 58 = 0	REG 77 = 16
REG 59 = 0	REG 78 = 15
REG 5a = 70	REG 79 = 16
REG 5b = 7d	REG 7a = 18
REG 5c = af	REG 7b = 38
REG 5d = 2d	REG 7c = 40
REG 5e = 3f	REG 7d = e0
REG 5f = 3f	


TTX-generator for the SAA7182/3 (EURO-DENC)

PAGE AND LINE ORGANIZATION FOR TELETEXT


A Teletext page consists of 24 rows of 45 bytes. The first row, the header or Row 0 is the leader row for the rest of the page. (Rows 1–23). The beginning (Byte 1–5) of each row has the same format, The first 3 bytes are used for bit and byte synchronization,

where the first two bytes are the run-in-code (Bit synchronizing) and the 3rd byte the Start code (Byte synchronizing). Bytes 4 and 5 are Hamming-Coded with the 3-Bit Magazine number and a 5-Bit Row address.

Prefix:



Magazine and Row address:



Hamming Codes:

DECIMAL	HAMMING	DECIMAL	HAMMING
0	15	8	D0
1	02	9	C7
2	49	10	8C
3	5E	11	9B
4	64	12	A1
5	73	13	B6
6	38	14	FD
7	2F	15	EA

- Example 1: Page 123 Row 0
hamming code 0215
- Example 2: Page 123 Row 1
hamming code C715
- Example 3: Page 123 Row 2
hamming code 0202
- Example 4: Page 555 Row 15
hamming code B62F

TTX-generator for the SAA7182/3 (EURO-DENC)

BIT SEQUENCE IN THE FIFO

Within 52µs 45 Bytes (= 360 Bits) are written, giving a data rate of 6,923 Mbit/s. The data rate is however 6,9375 Mbit/s, which is 444 times the television line frequency of 15625 Hz. As the clock frequency of the EURO-DENC is 27 MHz, a frequency conversion is required. This can be achieved by using the conversion table shown below.

$$f = 6,9375 \text{ MHz} * 144/37 = 27 \text{ MHz}$$

BIT/CLOCK

1/1	7/25	113/49	19/73	25/97	31/121
1/2	7/26	13/50	19/74	25/98	32/122
1/3	7/27	13/51	20/75	26/99	32/123
1/4	7/28	14/52	20/76	26/100	32/124
2/5	8/29	14/53	20/77	26/101	32/125
2/6	8/30	14/54	20/78	26/102	33/126
2/7	8/31	14/55	21/79	27/103	33/127
2/8	8/32	15/56	21/80	27/104	33/128
3/9	9/33	15/57	21/81	27/105	33/129
3/10	9/34	15/58	21/82	27/106	34/130
3/11	9/35	15/58	22/83	28/107	34/131
3/12	9/36	16/60	22/84	28/108	34/132
4/13	10/37	16/61	22/85	28/109	34/133
4/14	10/38	16/62	22/86	29/110	35/134
4/15	10/39	16/63	23/87	29/111	35/135
4/16	11/40	17/64	23/88	29/112	35/136
5/17	11/41	17/65	23/89	29/113	35/137
5/18	11/42	17/66	23/90	30/114	36/138
5/19	11/43	17/67	24/91	30/115	36/139
5/20	12/44	18/68	24/92	30/116	36/140
6/21	12/45	18/69	24/93	30/117	36/141
6/22	12/46	18/70	24/94	31/118	36/142
6/23	12/47	18/71	25/95	31/119	37/143
6/24	13/48	19/72	25/96	31/120	37/144

CHARACTER-CODE TABLE

Character	Code (Hex)	Character	Code (Hex)
A	c1	a	61
B	c2	b	62
C	43	c	e3
D	c4	d	64
E	45	e	e5
F	46	f	e6
G	c7	g	67
H	c8	h	68
I	49	i	e9
J	4a	j	ea
K	cb	k	6b
L	4c	l	ec
M	cd	m	6d
N	ce	n	6e
O	4f	o	ef
P	d0	p	70
Q	51	q	f1
R	52	r	f2
S	d3	s	73
T	54	t	f4
U	d5	u	75
V	d6	v	76
W	57	w	f7
X	58	x	f8
Y	d9	y	79
Z	da	z	7a
1	31	□	20
2	32	,	2c
3	b3	.	ae
4	b4	/	2f
5	b5	(	a8
6	b6	)	29
7	37	'	a7
8	38	*	2a
9	b9	—	ad
0	b0	+	ab
		;	bb

NOTES:

This Character-Code Table is only a part of the full Code-Table!
Bit 7 is the parity bit.

TTX-generator for the SAA7182/3 (EURO-DENC)

PART OF A TTX-PAGE- EPROM CODE

Page 333:

```

:100200005555275E155E5E1515151515D0C1C718      Line 0
      *1 ←      *2 ←      *3
:100210004520B3B3B3202020D3C1C120373138B338
:10022000202054454C4554455854202020000000BF
:1002300000000000000000000000000000000000BE
:100240005555279B15B0312020202020202020202C      Line 1
:10025000202020202020202020202020202020209E
:100260002020202020202020202020202020000000EE
:10027000000000000000000000000000000000007E
:100280005555275E02B03220202049CEC445582063      Line 2
:10029000202020D0524FCD20D6E5F273E9EF6E201A
:1002A00031AEB02020202020202020202020200000007F
:1002B0000000000000000000000000000000000003E
:1002C0005555279B02B0B3202020202020202020203D      Line 3
:1002D00020202020202020202020202020202020201E
:1002E000202020202020202020202020202000000006E
:1002F000000000000000000000000000000000000FE
:100300005555275E49B034202020D06167E5202074      Line 4
:1003100020202043EF6EF4E56EF4732020202020208F
:100320002020202020202020202020202000000002D
:1003300000000000000000000000000000000000BD
:100340005555279B49B0B52020202031B0B0202042      Line 5
:1003500020202020496E64E5F82020202020202045
:10036000202020202020202020202020200000000ED
:10037000000000000000000000000000000000007D
:100380005555275E5EB0B62020202031B0312020A8      Line 6
:1003900020202020436861F261E3F4E5F22073E558
:1003A000F420202020202020202020202020000000D9
:1003B000000000000000000000000000000000003D
:1003C0005555279B5EB0372020202031B0322020A9      Line 7
:1003D00020202020CEE5F773E6EC61736820202012
:1003E0002020202020202020202020202000000006D
:1003F00000000000000000000000000000000000FD
:100400005555275E64B0382020202031B0B320201D      Line 8
:100410002020202043ECEFE36B20E3F261E36BE567
:10042000F22020202020202020202020202000000005A
:1004300000000000000000000000000000000000BC
:100440005555279B64B0B920202020B5B5B5202094      Line 9
:1004500020202020D649C4454F5445585420202000
:10046000202020202020202020202020200000000EC
:10047000000000000000000000000000000000007C
:

```

NOTES:

*1 framing code (includes: clock run in)

*2 magazine and row address

*3 page
page

TTX-generator for the SAA7182/3 (EURO-DENC)

EXAMPLES OF TELETEXT PAGES:

P123 PAGE123 SAA7183 TELETEXT

SAA 7183 Philips
Semiconductors
DTV 1995

(EURO-DENC)

- Digital Pal/NTSC/SECAM encoder
- CVBS,Y,C and RGB available
- Internal Colour Bar Generator
- Closed Captioning and Teletext
Encoding
- Overlay with LUT's (8x3 bytes)
- Macrovision Pay-per-View System

P100 PAGE 100 SAA 7183 TELETEXT

01

02 PHILIPS MULTIMEDIA DTV 1995

03

04 Page	Contents	Page	Contents
05 100	Index new	333	Index old
05 101	Character set	123	EURO-DENC
07 102	News flash	156	TTX
08 103	Clock cracker		
09 555	VIDEOTEXT		

10

11

EURO DENC TEST

-

-

12

13

14

15

16

17

18

19 Conceal :

20

21

22 *** Large characters ***

23